NAG Fortran Library Routine Document

F04AMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AMF calculates the accurate least-squares solution of a set of m linear equations in n unknowns, $m \geq n$ and rank $= n$, with multiple right-hand sides, $AX = B$, using a QR factorization and iterative refinement.

2 Specification

```
 SUBROUTINE F04AMF(A, IA, X, IX, B, IB, M, N, IR, EPS, QR, IQR, ALPHA, E,
1                  Y, Z, R, IPIV, IFAIL)
 INTEGER           IA, IX, IB, M, N, IR, IQR, IPIV(N), IFAIL
 real              A(IA,N), X(IX,IR), B(IB,IR), EPS, QR(IQR,N), ALPHA(N),
1                  E(N), Y(N), Z(N), R(M)
```

3 Description

To compute the least-squares solution to a set of m linear equations in n unknowns $(m \geq n)AX = B$, this routine first computes a QR factorization of A with column pivoting, $AP = \mathrm{QR}$, where R is upper triangular, Q is an m by m orthogonal matrix, and P is a permutation matrix. Q^T is applied to the m by r right-hand side matrix B to give $C = Q^T B$, and the n by r solution matrix X is calculated, to a first approximation, by back-substitution in $RX = C$. The residual matrix $S = B - AX$ is calculated using ***additional precision***, and a correction D to X is computed as the least-squares solution to $AD = S$. X is replaced by $X + D$ and this iterative refinement of the solution is repeated until full machine accuracy has been obtained.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

1: A(IA,N) – ***real*** array *Input*

On entry: the m by n matrix A.

2: IA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AMF is called.

Constraint: IA $\geq$ M.

3: X(IX,IR) – ***real*** array *Output*

On exit: the n by r solution matrix X.

4: IX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F04AMF is called.

Constraint: IX $\geq$ N.

5: B(IB,IR) – ***real*** array *Input*

On entry: the m by r right-hand side matrix B.

6: IB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AMF is called.

Constraint: IB $\geq$ M.

7: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A, i.e., the number of equations.

8: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A, i.e., the number of unknowns.

Constraint: N $\leq$ M.

9: IR – INTEGER *Input*

On entry: r, the number of right-hand sides.

10: EPS – ***real*** *Input*

On entry: EPS must be set to the value of the ***machine precision***.

11: QR(IQR,N) – ***real*** array *Output*

On exit: details of the QR factorization.

12: IQR – INTEGER *Input*

On entry: the first dimension of the array QR as declared in the (sub)program from which F04AMF is called.

Constraint: IQR $\geq$ M.

13: ALPHA(N) – ***real*** array *Output*

On exit: the diagonal elements of the upper triangular matrix R.

14: E(N) – ***real*** array *Workspace*

15: Y(N) – ***real*** array *Workspace*

16: Z(N) – ***real*** array *Workspace*

17: R(M) – ***real*** array *Workspace*

18: IPIV(N) – INTEGER array *Output*

On exit: details of the column interchanges.

19: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The rank of A is less than n; the problem does not have a unique solution.

IFAIL $= 2$

The iterative refinement fails to converge, i.e., the matrix A is too ill-conditioned.

7 Accuracy

Although the correction process is continued until the solution has converged to full machine accuracy, all the figures in the final solution may not be correct since the correction D to X is itself the solution to a linear least-squares problem. For a detailed error analysis see page 116 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by the routine is approximately proportional to $n^2(3m - n)$, provided r is small compared with n.

9 Example

To calculate the accurate least-squares solution of the equations

$$1.1x_1 + 0.9x_2 = 2.2$$
$$1.2x_1 + 1.0x_2 = 2.3$$
$$1.0x_1 + 1.0x_2 = 2.1$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F04AMF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          MMAX, NMAX, IR, IA, IX, IB, IQR
      PARAMETER        (MMAX=8,NMAX=MMAX,IR=1,IA=MMAX,IX=NMAX,IB=MMAX,
     +                 IQR=MMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             EPS
      INTEGER          I, IFAIL, J, M, N
*     .. Local Arrays ..
      real             A(IA,NMAX), ALPHA(NMAX), B(IB,IR), E(NMAX),
     +                 QR(IQR,NMAX), R(MMAX), X(IX,IR), Y(NMAX), Z(NMAX)
      INTEGER          IPIV(NMAX)
*     .. External Functions ..
      real             X02AJF
```

```
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         F04AMF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04AMF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF (M.GT.0 .AND. M.LE.MMAX .AND. N.GT.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),(B(I,J),J=1,IR),I=1,M)
         EPS = X02AJF()
         IFAIL = 1
*
         CALL F04AMF(A,IA,X,IX,B,IB,M,N,IR,EPS,QR,IQR,ALPHA,E,Y,Z,R,
     +              IPIV,IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99998) 'Error in F04AMF. IFAIL =', IFAIL
         ELSE
            WRITE (NOUT,*) ' Solution'
            DO 20 I = 1, N
               WRITE (NOUT,99999) (X(I,J),J=1,IR)
   20       CONTINUE
         END IF
      ELSE
         WRITE (NOUT,99998) 'M or N is out of range: M =', M, ' N =', N
      END IF
      STOP
*
99999 FORMAT (1X,8F9.4)
99998 FORMAT (1X,A,I5,A,I5)
      END
```

9.2 Program Data

```
F04AMF Example Program Data
  3  2
  1.1  0.9    2.2
  1.2  1.0    2.3
  1.0  1.0    2.1
```

9.3 Program Results

```
 F04AMF Example Program Results

  Solution
    1.3010
    0.7935
```